Assignment - 2
Verilog Modelling Techniques

Department of Electronics and Communication Engineering
Indian Institute of Technology, Roorkee

ECN 104 Digital Logic Design

Contents

1 Introduction

Verilog supports a wide variety of modelling techniques. Different modelling techniques allow writing hardware description at different levels of abstraction, starting from the switch level modelling (PMOS/NMOS) and all the way up to behavioural modelling (algorithmic description). Each one of them has their own benefits and use cases. In this assignment we will discuss three of them, namely Gate-level modelling, structural modelling and behavioural modelling.

2 Gate-level modelling

Gate-level modelling in Verilog is used to describe a circuit only using logic gates. This approach is used to describe critical parts of a design, like adders and multipliers. Using gate-level implementation allows greater control over the design than other techniques. Gate-level modelling is only used for small scale design. Due to its complexity other modelling techniques are commonly used to abstract gate-level implementations.

2.1 Gate Primitives in Verilog

Verilog support following gates:

- AND
- NAND
- OR
- NOR
- XOR
- XNOR
- NOT

Gates in Verilog are available as primitives and can be instantiated similar to modules. Listing 1 is an example gate level circuit.

Listing 1: Example module using Gate-level modelling

```
/* Example using gate-level modelling, gate is a built-in
 * primitive in Verilog. Gates are instantiated in a way
 * similar to modules. Gates can be of single input or
 * multiple input
 */
module gateLevelExample(input1, input2, input3, result);
   input input1, input2, input3;
   output [8:0] result;

   // Single input gates
   not  g0(result[0], input1);           // = ~input1

   // Two input gates
   and  g1(result[1], input1, input2);   // =   input1 & input2
   nand g2(result[2], input1, input2);   // = ~(input1 & input2)
   or   g3(result[3], input1, input2);   // =   input1 | input2
   nor  g4(result[4], input1, input2);   // = ~(input1 | input2 )
   xor  g5(result[5], input1, input2);   // =   input1 ^ input2
   xnor g6(result[6], input1, input2);   // = ~(input1 ^ input2)

   // Gates with more than two input
   xnor g7(result[7], input1, input2, input3);
   and  g8(result[8], input1, input2, input3, input1);
endmodule // gateLevelExample
```

Verilog also supports instantiating gates without a instance name demonstrated in Listing 2.

Listing 2: Instantiating unnamed gates

```
/*
 * Gates in Verilog can be instantiated without a name,
 * such instantiation in Verilog is legal.
 */
module unnamedGate(input1, input2, result);
   input input1, input2;
   output result;

   and(result, input1, input2); // Unnamed gate
endmodule // unnamedGate
```

2.2 Delay specification

All the circuits we have studied so far have no delay associated with them, these are called 0-delay circuits. Real circuits however, always have a delay between their input and output. Verilog allows modelling of delays at various level of abstraction using delay statements.

Syntax for specifying a delay is:

```
<gate_primitive> #(<delay>) <inst_name>(...ports...)
```

For example:

```
/* 2-input AND gate with 1 time unit delay */
and #(1) a1(result, input1, input2);
```

To specify unit of delay, we'll write a compiler directive. This is typically written at the begining of the description.

```
`timescale 1ns/1ps
```

Here, 1ns is the unit time while 1ps is the time resolution which will be explained in the assignment. A complete example using delay statements is given in Listing 3.

Listing 3: Example usage of delays statement to specify propagation delay of logic gates.

```
1  /* Compiler directive to specify time unit, which will be
2   * used for assigning propogation delays of logic gates.
3   */
4  `timescale 1ns/1ps
5
6  module delayExample(input1, input2, result);
7     input input1, input2;
8     output result;
9
10    // Specify a NAND gate having propogation delay of 2ns
11    nand #(2) nd1(result, input1, input2);
12 endmodule // delayExample
```

Problem 1

NOT gate is one of the primitives available with Verilog. NOT gate is a type of buffer and is also known as inverting buffer. Buffers in digital circuit are often used to isolate two parts of a circuit, but this isolation can also be added with a certain behaviour like inversion of signal or delaying signal to synchronize with other parts.

1. Write a Verilog module which acts as an inverting buffer (that is, a NOT gate), you have to use verilog primitive `not` for the implementation.

2. Another kind of buffer often used in digital circuit is the 'non-inverting buffer'. As the name suggests, the non-inverting buffer doesn't invert the input unlike the inverting buffer. This buffer seems to be of no use since it cannot do any processing. This leads us to its one of the common use case, using buffer to level translation. Level translation is used whenever two parts of a circuit are powered with different Vdd (different supply voltages). Now that you know what a buffer is, write a Verilog module to implement a buffer using the inverting buffers you designed in Problem 1.1.

3. NOT gates are also used in ring oscillators, which is a kind of an oscillator often used to generate clock signals. Ring oscillator is an oscillator which looks like a ring, and contains a chain of odd number of NOT gates. Ring oscillator's frequency depends on the delay of the NOT gate and number of gates used. A simple 3 NOT gate ring oscillator is shown in Fig. 1. Let the propagation delay of each of the gate in Fig. 1 be t_d, intuitively one can say that the delay between change of values at any point is $3t_d$. Use this information and calculate the time period of an n NOT gate ring oscillator, where n is an odd number.



Figure 1: Simulation result for Listing 1, output changes 15ns after the change in input.

4. Using the expression from the Problem 1.3, write the description of a module for a 5 NOT gate ring oscillator. You have to design the ring oscillator such that the time period of the generated wave will be 30ns, assume that each of the NOT gate has a propagation delay of 5ns.

5. Will a single not gate ring oscillator work? Why or why not?

---

3 Data-flow Modelling

Data flow modelling is a higher level of abstraction than the gate level modelling we just studied. Describing a circuit using data-flow modelling does not require knowledge of gates level circuit, thus it is easier than gate-level modelling when description of large scale circuits are written. All the examples from Assignment 1 used data flow modelling.

3.1 Continuous Assignment

Continuous assignment in Verilog is used for data flow modelling. These assignment starts with `assign` keyword. Continuous assignment drives value into a net (`wire`). Following example describes use of continuous assignment:

Note: Verilog is concurrent language unlike programming languages such as C,C++ or Java. All the continuous assignments are evaluated at the same time.

Listing 4: Example usage of continuous assignment.

```
/*
 * Following module uses continuous assignement to model
 * an AND gate
 */
module continuousAssignment(input1, input2, result);
   input input1, input2;
   output result;

   /* Use continuous assignment to set result
    * This is equalent to:
    *     and (result, input1, input2);
    */
   assign result = input1 & input2;
endmodule // continuousAssignment
```

Continuous assigment in Verilog can also be done implicitly, which is assigning value on declaration of a net (`wire`). Implicit delcaration of Verilog is:

```
wire new_wire = input1 & input2;
```

3.2 Assignment Delays

Similar to gate-level modelling, Verilog allows specifying delays in assignment to model real circuits. Assignment delay specify the delay between the change of LHS and RHS of a continuous assignment. Listing 5 shows example usage of assignment delay while Fig. 2 shows simulation result of Listing 5.

Listing 5: Using assignment delay in Verilog.

```
`timescale 1ns/1ps

module assignmentDelay(input1, input2, result);
   input input1, input2;
   output result;

   // Defines a circuit which will have output value
   // input1 | input2 15ns after changing input
   assign #15 result = input1 | input2;
endmodule // assignmentDelay
```



Figure 2: Simulation result for Listing 5, output changes 15ns after the change in input.

Problem 2

Multiplexer is a digital element which is used to select a single signal from a group of signals. Block diagram of a multiplexer is shown in Fig. 3, the input in_1 and in_2 are multiplexed and the output is decided using the input c. If the input c is 0 then output $= in_1$ else output $= in_2$.



Figure 3

1. Write the hardware description of a 2-to-1 multiplexer using Verilog. You have to implement this using only data-flow modelling. (*Hint: Verilog supports the ternary operator, which has same behaviour as in programming languages.*)
2. Now suppose you want to model a multiplexer you just purchased from the market, which has a propagation delay of 2ns. Modify the module from part 1 such that the behaviour of your description matches the one you bought.

---

4 Behavioral Modelling

Behavioral modelling is an even higher level of modelling where circuit description is written as its behaviour, this algorithmic representation of a circuit abstracts the details of gate-level and data flow modelling. Behavioral modelling resembles more to programming languages such as C than it does to circuit description.

4.1 `reg` Element

`reg` element is used to represent abstract storage device in Verilog. `reg`s can be used to store information (single bit or of arbitrary length using array, arrays are similar to vectors). We'll get back to usage of `reg` and will differentiate it with `wire` after studying procedural blocks.

4.2 Procedural Blocks

Earlier we read about continuous assignment which allows us to drive value to a net whenever the driver changes. This kind of assignment allows only to describe combinational circuit. To describe sequential circuits Verilog provides procedural blocks. These blocks are used to drive values to net (`reg`) only if the condition is met.

4.3 `initial` Block

`Initial` blocks in Verilog are used to specify initial values of all the storage elements. When simulation starts simulator doesn't know what values are to be assigned to storage elements. Initial block begins with `initial begin` and ends with `end`. Initial blocks gets executed only once when the simulation is started.

Note: Only a `reg` can be assigned values inside an `initial` block. This is because unlike `wire` which are used for connection, `reg` stores information and this information is unknown to the simulator at t=0. Structure of an initial block is given in Listing 6.

Listing 6: Structure of an initial block.

```
...
  initial begin
      ...
      // Specify initial condition
      ...
  end
...
```

4.4 `always@` block

`always@` block in Verilog are used for describing an event which should happen only under certain conditions, such as change in value of one of the elements.

Basic structure of an `always@` block is given in Listing 7.

Listing 7: Structure of an always@ block.

```
...
  always @( ...condition... ) begin
      ...
      // Description of conditional event
      ...
  end
...
```

Listing 8 shows a module which changes output only one the positive edge of input `clk`.

Listing 8: Synchronous logic which changes value of result only at the positive edge of clk.

```
/*
 Synchronous module which changes output only on positive
 edge of clk.
 */
module alwaysExample(clk, input1, input2, result);
   input clk, input1, input2;
   output result;

   // Declare output as reg, this will be explained later
   reg result;

   // Here condition is *positive edge of clk*, this implies
   // following event only takes place when a positive edge
   // of clk arrives
   always @(posedge clk) begin
```

```
16          // Assign value to reg result using procedural assignment
17          result = input1 & input2;
18      end
19  endmodule // alwaysExample
```

4.5 Blocking and Non-Blocking Assignment

4.5.1 Blocking Assignments

Blocking assingments are assignments which block the simulation while their value is being calculated. This means that the execution flow will stop at this statement until it is executed. Blocking assignment in Verilog is done using =. Listing 9 demonstrates the functioning of blocking assignment.

Listing 9: Swapping bytes using blocking assignment.

```
1  // Following example is for swapping bytes
2  // input is of size 16 bits (2 bytes)
3  // This example does not work due to the use of blocking
4  // assignment.
5
6  // This example is taken from:
7  // http://www.sutherland-hdl.com/papers/1996-CUG-presentation_nonblocking_assigns.pdf
8
9  module swapBytes;
10     ...
11     reg [15:0] temp;
12     ...
13
14     // Can you guess the reason why this block won't swap
15     // bytes?
16     always @(...some_condition...) begin
17         temp[15:8] = temp[7:0];
18         temp[7:0] = temp[15:8];
19     end
20 endmodule // swapBytes
```

In the above example (Listing 9) statement 18 doesn't gets executed until statement 17 is executed, this is due to the use of blocking assignment.

4.5.2 Non-blocking Assignments

In Listing 9 we saw that blocking assignments cannot be used to swap bytes, this is where non-blocking assignments will come to use. Non-blocking assignments are evaluated in two steps first all the RHS values are calculated at the begining of the procedural block, and then the value is assigned to LHS when the execution reaches particular statement.

Listing 10 shows how non-blocking assignments can be used for swapping bytes.

Listing 10: Swapping bytes using non-blocking assignment it works!

```
1  // Following example is for swapping bytes
2  // using non-blocking assigment, this module
3  // unlike last example uses non-blocking assignment
4  // this allows swapping bytes correctly.
5
6  // This example is taken from:
7  // http://www.sutherland-hdl.com/papers/1996-CUG-presentation_nonblocking_assigns.pdf
8
9  module swapBytes;
10     ...
11     reg [15:0] temp;
12     ...
13
```

```
14      // Can you guess the reason why this block won't swap
15      // bytes?
16      always @(...some_condition...) begin
17          temp[15:8] <= temp[7:0];
18          temp[7:0] <= temp[15:8];
19      end
20  endmodule // swapBytes
```

5 Differences between `wire` and `reg` and where to use what

5.1 Legal use of `wire`

Wires in Verilog are used to connect two elements. They can be assigned a value or a value can be read from them. However they cannot store it. You'll have to drive them with values (constant, other wires or regs).

- `wire`s are allowed only in continuous assignments (page 4).
- A `wire` cannot be assigned a value inside a procedural block.
- A `wire` can be used to assign a value to a `reg` or a `wire`.
- `wire`s can be used for I/O connections of a module instance.

5.2 Legal use of `reg`

`reg`s in Verilog are storage elements. They, however, do not represent physical registers. Once synthesized they can be represented by a physical register, RAM or ROM.

- `reg`s cannot be assigned a value using continuous assigment.
- A `reg` can only be assigned a value in a procedural block.
- A `reg` can be used to assign a value to a `reg` or a `wire`.

5.3 Places where both `wire` and `reg` are allowed

- Both can appear on the right hand side of an `assign` statement or can be used inside a procedural block (`initial`/`always`) block to set value of a `reg`.
- Can be used as inputs to a module.

Problem 3

A Turing machine is a theoretical machine which is used as an abstraction to real computational machines, Turing machine was proposed by Alan Turing in 1936. The basis of invention of turing machine was Alan's interest in the philosophy of computation. An interesting thing about the Turing machine is that any algorithm which is impossible to implement on Turing machine has never worked on any other real machine. Programming languages such as Java, C, C++ etc. are said to be turing complete This implies that anything that can be computed by a turing machine can be computed by these languages.

Basics of Turing machine

Turing machine consist of an infinitely long tape which has certain shapes (or characters) on it, this tape is traversed by a head which reads the character directly under it and takes appropriate movement decision. An example of such machine is shown in Fig. 4.

The video at https://www.youtube.com/watch?v=mPec64RUCsk will help you understand the Turing machine a bit better.

Turing machines can be used to implement Finite state machines. To use a Turing machine as a finite state machine we need following assumptions:



Figure 4: Example of a Turing machine.

Value under head	Direction of movement	Write value
A + _	Right Right Left	A

(a) The tape is finite in length (since it would be impossible to write description of an infinite tape).

(b) Each character on the tape can only be 'A', '+' or '_' where 'A' and '+' are used for decisions while '_' represents an unused tape slot.

(c) The input to the machine will be the initial configuration of the tape while the output of the machine will be the state of the tape when the machine halts.

(d) At every clock cycle the Turing machine can execute only one instruction.

(e) Here each possible arrangement of characters on the tape represents a single state in the FSM.

Now consider the above mentioned Turing machine rules for a machine which adds up two numbers. These numbers are represented in base 1 (e.g. 1111_1 to represent 4_{10}):

Using this rule set addition of two numbers can be performed on a Turing machine, say for example you want to add 2_{10} (AA_1) and 4_{10} ($AAAA_1$) the trace of the Turing machine to perform such addition would be (Assuming the halting condition to be a state where head returns to the position one left to the first number):

1. __[A]A+AAAA___
2. __A[A]+AAAA___
3. __AA[+]AAAA___
4. __AAA[A]AAA___
5. __AAAA[A]AA___
6. __AAAAA[A]A___
7. __AAAAAA[A]___
8. __AAAAA[A]____
9. __AAAA[A]A____
10. __AAA[A]AA____
11. __AA[A]AAA____
12. __A[A]AAAA____
13. __[A]AAAAA____
14. _[_]AAAAAA____ (**HALT**)

Here [.] represents the head of the machine.

Questions

1. Briefly describe the algorithm which would give the above trace (you can use psuedo code for your description of the algorithm).
2. Write Verilog description for the Turing machine described above, use 2 bit registers to represent a single location on the tape. The tape of the machine will be finite in length and should be enough to store the above example.

3. Simulate the addition of 2 and 4 in the above turing machine.

If you want to read more interesting stuff on Turing Machines, you can look up for ‘The halting problem’ on the internet.

References

- http://inst.eecs.berkeley.edu/~cs150/fa08/Documents/Always.pdf
- Aenean in sem ac leo mollis blandit.

Additional resources

- Stack Exchange: How are Verilog “always” statements implemented in hardware?
  http://bit.ly/verilogAlwaysStatement
- Stack Overflow: Difference between behavioral and dataflow in verilog
  http://bit.ly/modellingDiff
- Stack Overflow: How can I know if my code is synthesizable? [Verilog]
  http://bit.ly/synthCode

These resources are great for expanding your understanding, but...